

Example: Exercise 11, Section 16.8 (Stokes' Theorem)

We want to use Maple to verify Stokes' Theorem, and also to do some graphical work.

> #Define the vector field:

> with(VectorCalculus): with(Student[VectorCalculus]):

> F:=VectorField(<x^2*z, x*y^2, z^2>);

$$F := (x^2 z) \bar{e}_x + (x y^2) \bar{e}_y + (z^2) \bar{e}_z \quad (1)$$

> #Define the surface:

> S:=<u*cos(v),u*sin(v), 1-u*cos(v)-u*sin(v)>; # u is between 0 and 3, v is 0 to 2*Pi

$$S := (u \cos(v)) e_x + (u \sin(v)) e_y + (1 - u \cos(v) - u \sin(v)) e_z \quad (2)$$

> C:=<3*cos(t), 3*sin(t), 1-3*cos(t)-3*sin(t)>;

$$C := 3 \cos(t) e_x + 3 \sin(t) e_y + (1 - 3 \cos(t) - 3 \sin(t)) e_z \quad (3)$$

> #First, the line integral:

> LineInt(F, Path(C, t=0..2*Pi));

$$\frac{81}{2} \pi \quad (4)$$

> # The line integral by hand:

> Integrand:=simplify(DotProduct(subs({x=C[1],y=C[2],z=C[3]},F), diff(C,t)));

$$\begin{aligned} \text{Integrand} := & -162 \cos(t)^4 + 81 \cos(t)^3 \sin(t) - 54 \cos(t)^3 - 81 \cos(t)^2 \sin(t) \\ & + 198 \cos(t)^2 + 24 \cos(t) + 30 \sin(t) - 18 \end{aligned} \quad (5)$$

> int(Integrand,t=0..2*Pi);

$$\frac{81}{2} \pi \quad (6)$$

> # Now the surface integral using Stokes' Theorem:

> G:=Curl(F);

$$G := (x^2) \bar{e}_y + (y^2) \bar{e}_z \quad (7)$$

> N:=CrossProduct(diff(S,u),diff(S,v));

$$\begin{aligned} N := & (\sin(v) (u \sin(v) - u \cos(v)) - (-\cos(v) - \sin(v)) u \cos(v)) e_x + \\ & (-\cos(v) (u \sin(v) - u \cos(v)) - (-\cos(v) - \sin(v)) u \sin(v)) e_y + (\cos(v)^2 u \\ & + \sin(v)^2 u) e_z \end{aligned} \quad (8)$$

> Integrand:=DotProduct(subs({x=S[1],y=S[2],z=S[3]},G),N);

$$\begin{aligned} \text{Integrand} := & u^2 \cos(v)^2 (-\cos(v) (u \sin(v) - u \cos(v)) - (-\cos(v) \\ & - \sin(v)) u \sin(v)) + u^2 \sin(v)^2 (\cos(v)^2 u + \sin(v)^2 u) \end{aligned} \quad (9)$$

> int(int(Integrand,u=0..3),v=0..2*Pi);

$$(10)$$

$$\frac{81}{2} \pi$$

(10)

> #Graph the surface and the curve C:

> with(plots):

> A:=spacecurve(C,t=0..2*Pi,color=red,thickness=3);

A:= PLOT3D(...)

(11)

> B:=plot3d(S,u=0..3,v=0..2*Pi);

B:= PLOT3D(...)

(12)

> BB:=fieldplot3d(F,x=-3..3,y=-3..3,z=-3..5,fieldstrength=fixed,
color=black);

BB:= PLOT3D(...)

(13)

> display3d(A,B,BB,scaling=constrained);

